// File: lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: src/AIWorldsContribution.sol



pragma solidity ^0.8.19;

import {Ownable} from "@openzeppelin/access/Ownable.sol";
import {IERC20} from "@openzeppelin/token/ERC20/IERC20.sol";

contract AIWorldsContribution is Ownable {
    struct Contribution {
        address addr;
        uint256 amount;
    }
    IERC20 public token;
    uint256 public constant MIN_CONTRIBUTION = 0.1 ether;
    uint256 public constant MAX_CONTRIBUTION = 5 ether;
    uint256 public supply;
    uint256 public capacity;
    uint256 public contributorCapacity;
    bool public contributionOpen;
    uint256 public totalContributed;
    uint256 public numContributors;
    mapping(uint256 => Contribution) public contribution;
    mapping(address => uint256) public contributor;

    constructor(
        uint256 capacity_,
        uint256 contributorCapacity_
    ) {
        capacity = capacity_;
        contributorCapacity = contributorCapacity_;
    }

    /**
     * @notice Contribute to AIWorlds (Min 0.1 ETH, Max 5 ETH)
     */
    function contribute() public payable {
        require(contributionOpen, "Contributions Not Open");
        uint256 currentContribution = contribution[contributor[msg.sender]]
            .amount;
        require(msg.value >= MIN_CONTRIBUTION, "Under Minimum Contribution");
        require(
            msg.value + currentContribution <= MAX_CONTRIBUTION,
            "Over Maximum Contribution"
        );
        totalContributed += msg.value;
        require(totalContributed <= capacity, "Capacity Reached");
        uint256 contributionIndex;
        if (contributor[msg.sender] != 0) {
            contributionIndex = contributor[msg.sender];
        } else {
            contributionIndex = numContributors + 1;
            require(
                contributionIndex <= contributorCapacity,
                "Contributor Capacity Reached"
            );
            numContributors++;
        }
        contributor[msg.sender] = contributionIndex;
        contribution[contributionIndex].addr = msg.sender;
        contribution[contributionIndex].amount += msg.value;
    }

    function airdrop() external onlyOwner {
        _airdrop(1, numContributors);
    }

    function airdropex(uint256 from, uint256 to) external onlyOwner {
        _airdrop(from, to);
    }

    function _airdrop(uint256 from, uint256 to) private {
        require(address(token) != address(0), "Unset Token");
        require(supply != 0, "Unset Supply");
        uint256 pricePerToken = (totalContributed * 10e18) / supply;
        for (uint256 i = from; i <= to; i++) {
            uint256 contributionAmount = contribution[i].amount * 10e18;
            uint256 numberOfTokensToMint = contributionAmount / pricePerToken;
            token.transfer(contribution[i].addr, numberOfTokensToMint);
        }
    }

    function open() external onlyOwner {
        contributionOpen = !contributionOpen;
    }

    function refund() external onlyOwner {
        _refund(1, numContributors);
    }

    function refundex(uint256 from, uint256 to) external onlyOwner {
        _refund(from, to);
    }

    function _refund(uint256 from, uint256 to) private {
        for (uint256 i = from; i <= to; i++) {
            address payable refundAddress = payable(contribution[i].addr);
            refundAddress.transfer(contribution[i].amount);
        }
    }

    function withdraw() external onlyOwner {
        (bool success, ) = payable(owner()).call{value: address(this).balance}(
            ""
        );
        require(success, "Withdraw failed");
    }

    function withdrawERC20() external onlyOwner {
        uint256 balance = token.balanceOf(address(this));
        if (balance > 0) {
            token.transfer(owner(), balance);
        }
    }

    function setToken(IERC20 token_) external onlyOwner {
        token = token_;
    }

    function setSupply(uint256 supply_) external onlyOwner {
        supply = supply_;
    }

    function reduceCapacity(uint256 capacity_) external onlyOwner {
        require(capacity_ <= capacity, "Reduce Only");
        capacity = capacity_;
    }

    function reduceContributorCapacity(
        uint256 contributorCapacity_
    ) external onlyOwner {
        require(contributorCapacity_ <= contributorCapacity, "Reduce Only");
        contributorCapacity = contributorCapacity_;
    }
}